Analyzing and Visualizing WRF-ARW data using NCL



18th Annual WRF User's Workshop

Mary Haley • CISL / TDD / VAST

June 16, 2017

The National Center for Atmospheric Research is sponsored by the National Science Foundation

Goals for this 90-minute tutorial

- Brief introduction to NCL and WRF-NCL
- Demo three ways for plotting WRF-ARW data
- Provide **TIPS** along the way
- Give you time to try sample scripts

Bookmark this page

http://www.ncl.ucar.edu/Training/Tutorials/WRF_Users_Workshop/

A scripting language developed at NCAR and tailored for the analysis and visualization of geoscientific data



http://www.ncl.ucar.edu/

NCL

NCAR Command Language

NCL Overview

- A scripting language similar to Matlab, Python, IDL
- Tailored to climate and weather
- Has variable types, if-then-end if, do loops, arithmetic operators, functions, procedures
- Built around the NetCDF variable model
- F90-like array arithmetic
- Can call your own Fortran or C routines



Why use NCL?

- Developed in NCAR/CISL in close collaboration with NCAR scientific staff
- Mature package (20+ years)
- Open source, free
- Extensive website, hundreds of examples
- Well-supported
- Intensive training workshops



1. File input and output

- Data model based on netCDF model (metadata describes data)
- **One function** reads all supported data formats:
  - NetCDF3, GRIB 1 and 2, HDF4, HDF5, HDF-EOS2, HDF-EOS5, shapefiles, NetCDF4
  - Writes NetCDF3, NetCDF4, and HDF4
- OPeNDAP-enabled client available
- ASCII, Fortran/C binary (read and write)
- “Never fear a data format”



2. Data analysis

- Array-based math
- Hundreds of functions
  - WRF-ARW specific functions
  - Climatologies
  - Spherical harmonics
  - Interpolation and regridding
  - Crop / Heat stress
  - EOFs
- Most automatically handle missing data





3. Visualization



- Publication quality and customizable 2D visualizations
- Contours, XY, vectors, wind barbs, streamlines
- Maps with common map projections
- Handles data on rectilinear, curvilinear, and unstructured grids (MPAS, triangular meshes)
- Specialized scripts for meteograms, skew-T, wind roses, histograms, cross section, panels
- Over 1,400 visualization “options”













WRF-NCL

*NCL suite of **analysis** and **visualization** functions tailored for WRF-ARW model data*

WRF-NCL Overview

- **Included with NCL since 2006**
- Developed by staff in NCAR/MMM
- Maintained by MMM and CISL
- Functions for calculating basic diagnostics (**`wrf_user_getvar`**)
- Functions for customized visualizations
- Website with lots of analysis and visualization examples



Full list of WRF-NCL functions

http://www.ncl.ucar.edu/Document/Functions/wrf.shtml

NCL functions to be called by user

wrf_user_getvar
wrf_user_ij_to_ll
wrf_user_intrp2d
wrf_user_intrp3d
wrf_user_list_times
wrf_user_ll_to_ij
wrf_user_unstagger
wrf_user_vert_interp

Graphics

wrf_contour
wrf_vector
wrf_overlays
wrf_map
wrf_map_overlays
wrf_map_resources
wrf_map_zoom

WPS intermediate files

wrf_wps_open_int
wrf_wps_rddata_int
wrf_wps_rdhead_int
wrf_wps_read_int
wrf_wps_write_int
wrf_wps_close_int

Computational routines not generally called directly by user

wrf_avo	wrf_pvo	wrf_smooth_2d	wrf_interp_1d
wrf_cape_2d	wrf_rh	wrf_helicity	wrf_interp_2d_xy
wrf_cape_3d	wrf_slp	wrf_updraft_helicity	wrf_interp_3d_z
wrf_dbz	wrf_uvmet	wrf_virtual_temp	wrf_ij_to_ll
wrf_eth	wrf_td	wrf_wetbulb	wrf_ll_to_ij
wrf_omega	wrf_tk		

Main WRF-NCL function: **wrf_user_getvar**

wrf_user_getvar - Get fields from input file and/or calculate diagnostics

```
a = addfile("wrfout_d01_2005-08-28_00:00:00.nc","r")

cttmp = wrf_user_getvar(a,"ctt",0) ;  0 → first time step

 slp  = wrf_user_getvar(a,"slp",1) ;  1 → second time step

  tc = wrf_user_getvar(a,"tc",-1) ; -1 -> all time steps

 hgt = wrf_user_getvar(a,"ter",0) ; terrain, 1st time step
```

Main WRF-NCL function: **wrf_user_getvar**

avo	absolute vorticity [10-5 s-1]
eth	Equivalent Potential Temperature [K]
cape_2d	Returns 2D fields mcape/mcin/lcl/lfc
cape_3d	Returns 3D fields cape and cin
ctt	Cloud Top Temperature [degC]
dbz	Reflectivity [dBZ]
mdbz	Maximum reflectivity [dBZ]
geopt	Full model geopotential [m2 s-2]
helicity	Storm Relative Helicity [m-2/s-2]
omg	Omega [C]
p	Full model pressure [Pa]
pressure	Full model pressure [hPa]
pvo	potential vorticity [PVU]
pw	Precipitable Water
rh2	2m Relative Humidity [%]
rh	Relative Humidity [%]
slp	Sea level pressure [hPa]

ter	Model terrain height [m]
td2	2m dew point temperature [C]
td	Dew point temperature [C]
tc	Temperature [C]
theta	Potential temperature [K]
tk	Temperature [K]
tv	Virtual temperature [K]
twb	Wet bulb temperature [K]
updraft_helicity	Updraft helicity [m-2/s-2]
ua	U component of wind on mass points
va	V component of wind on mass points
wa	W component of wind on mass points
uvmet10	10m U and V components of wind rotated to earth coordinates
uvmet	U and V components of wind rotated to earth coordinates
z / height	Full model height [m]

Demo: wrf_user_getvar

Using wrf_user_getvar to read data

and / or calculate diagnostics

`wrf_demo_getvar_simple.ncl`

`wrf_demo_getvar_all.ncl`

`wrf_demo_getvar_clo.ncl`

Run NCL script with "-n" option to turn off "**(0)**" output.

TIP

```
ncl -n wrf_demo_getvar_simple.ncl
```

Use "-Q" to turn off version and copyright

TIP

```
ncl -n -Q wrf_demo_getvar_all.ncl
```

TIP Can set NCL variables when you run a script:

```
ncl 'd="slp"' wrf_demo_getvar_clo.ncl
```

```
f = addfile("wrfout_d01_2008-09-29_16:30:00","r")

var = wrf_user_getvar(f,d,0)

printVarSummary(var)
printMinMax(var,0)
```

wrf_demo_getvar_simple.ncl

```
;--- Open WRF output file
f = addfile("wrfout_d01_2008-09-29_16:30:00","r")

;--- Read HGT variable
hgt = wrf_user_getvar(f,"HGT",0)      ; read first time step

printVarSummary(hgt)                  ; summary of variable, no values!
printMinMax(hgt,0)                    ; min/max of variable
print(hgt)                            ; prints EVERYTHING, can be too much

;--- Calculate sea level pressure at all time steps
slp = wrf_user_getvar(f,"slp",-1) ; [Time | 1] x [south_north | 197] x [west_east | 206]
printVarSummary(slp)
printMinMax(slp,0)

;--- Calculate u and v on mass points; put on same grid
u = wrf_user_getvar(f,"ua",0)
v = wrf_user_getvar(f,"va",0)

printVarSummary(u)   ; [bottom_top | 32] x [south_north | 197] x [west_east | 206]
printVarSummary(v)   ; the same size as "u"
```

Three ways to plot WRF-ARW data

- WRF-specific functions
  - wrf_contour
  - wrf_vector
  - wrf_map_overlays / wrf_overlays
- gsn_csm functions (native and non-native map)
  - gsn_csm_contour_map
  - gsn_csm_vector_map

Demo: plotting “HGT” variable

Using WRF-NCL functions to create plots

`wrf_demo_plot_hgt.ncl`

`wrf_demo_plot_hgt_custom.ncl`

wrf_demo_plot_hgt.ncl

```
; This load no longer needed in NCL V6.4.0
; load "$NCARG_ROOT/lib/ncarg/nclscripts/wrf/WRFUserARW.ncl"

;--- Open file and read a variable
f = addfile("wrfout_d01_2005-08-28_00:00:00","r")
hgt = wrf_user_getvar(f,"HGT",0)

printVarSummary(hgt)    ; LOOK AT YOUR DATA!
printMinMax(hgt,0)

;--- Where to send the graphics
wks = gsn_open_wks("x11","wrf_demo_plot_hgt") ;"png","pdf""svg"

;--- Set one plotting resource
res = True
res@cnFillOn = True          ; Turn on color fill

;--- Create a contour plot
contour = wrf_contour(f,wks,hgt,res)

;--- Draw the contours over a map
plot = wrf_map_overlays(f,wks,contour,False,False)
```

Init: 2005-08-28_00:00:00



OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

wrf_demo_plot_hgt.ncl *with minor changes*

TIP *Setting resources to see map outlines better*

```
;--- Open file and read a variable
f = addfile("wrfout_d01_2005-08-28_00:00:00","r")
hgt = wrf_user_getvar(f,"HGT",0)

;--- Where to send the graphics
wks = gsn_open_wks("x11","wrf_demo_plot_hgt")

;--- Set one plotting resource
res = True
res@cnFillOn = True          ; Turn on color fill
contour = wrf_contour(f,wks,hgt,res)

;--- Draw the contours over a map
mpres = True
mpres@mpGeophysicalLineThicknessF = 2.0      ; WRF-NCL uses 0.5
mpres@mpGeophysicalLineColor      = “black” ; and “gray”

plot = wrf_map_overlays(f,wks,contour,False,mpres)
```

BEFORE

Init: 2005-08-28_00:00:00



OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

AFTER

Init: 2005-08-28_00:00:00



OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

Demo: overlaying multiple plots

Creating multiple overlays

`wrf_demo_plot_overlays.ncl`

wrf_demo_plot_overlays.ncl

```
f = addfile("wrfout_d01_2005-08-28_00:00:00","r")

;--- Open a file and get several diagnostics
slp = wrf_user_getvar(f, "slp", 0)
t2  = wrf_user_getvar(f, "T2",  0)
u10 = wrf_user_getvar(f, "U10", 0)
v10 = wrf_user_getvar(f, "V10", 0)

wks = gsn_open_wks("x11","wrf_demo_plot_overlays")

;--- Line contours
os                  = True
os@cnLineColor      = "NavyBlue"
os@cnLineThicknessF = 2.0
plt_slp             = wrf_contour(f,wks,slp,os)

;--- Filled contours
ot              = True
ot@cnFillOn     = True
plt_tc          = wrf_contour(f,wks,t2,ot)

;--- Vectors
ov              = True
ov@NumVectors   = 47
plt_vec         = wrf_vector(f,wks,u10,v10,ov)

;--- Overlay vectors, line contours, and filled contours on a map
plot = wrf_map_overlays(f,wks,(/plt_tc,plt_slp,plt_vec/),False,False)
```

Init: 2005-08-28_00:00:00



OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1


REAL-TIME WRF
MainTitle
Init: 2005-08-28_00:00:00
Valid: 2005-08-28_00:00:00
Init time title not modifiable
Turn off with InitTime=False
TEMP at 2 M (K)
SLP (hPa)
U at 10 M (m s-1)
FieldTitle (UnitsLabel)
FieldTitle (UnitsLabel)
FieldTitle (UnitsLabel)
ValidTime=False by default
Change with TimeLabel
36°N
34°N
32°N
30°N
28°N
26°N
24°N
22°N
20°N
100°W
95°W
90°W
85°W
80°W
1012
1008
1008
1008
1000
1004
996
988
1004
1004
1000
1008
FieldTitle (UnitsLabel)
SLP Contours: 984 to 1014 by 2
TEMP at 2 M (K)
FieldTitle (UnitsLabel)
290 292 294 296 298 300 302 304 306 308 310
Bottom title is not modifiable
Turn off with Footer=False or NoHeaderFooter=False
OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

Where to find WRF-NCL scripts

NCL Examples Page:
http://www.ncl.ucar.edu/Applications/

WRF-NCL Examples Page:
http://www.ncl.ucar.edu/Applications/wrf.shtml

http://www.mmm.ucar.edu/wrf/OnLineTutorial/Graphics/NCL/

WRF ARW OnLineTutorial

http://www.mmm.ucar.edu/wrf/OnLineTutorial/Graphics/NCL/

Scripts and full-sized images available.

Google
"WRF ARW NCL"

Basic Plots  Basic Plot Setup *(This series of examples takes users though same basic steps in generating plotting scripts.)* Get and plot a single field Multiple input files	**Basic Surface Plots**  Surface 1 Surface 3 Surface 2	**Plots on Model Levels**  Clouds Levels from wrfout files Levels from metgrid files	**Plots on Interpolated Levels**  Height Levels Pressure Levels
Plotting Precipitation  Precipitation	**Diagnostics**  CAPE dBZ Vorticity *(More diagnotics are available, shown are only some newer/special diagnostics)*	**Cross-section Plots**  Height - Through a Pivot Point Height - Point A to Point B Pressure Limited Vertical Extent For 2D fields	**Skew_T Plots**  Skew_T
Speciality Plots  Overlay Zoom Overlay & Zoom Panel 1 Panel 2 Meteograms WRF Time Series data All fields in a file	**Preview Domain**  *This functionality, although available in NCL version 5.0.1, is still experiential.* Preview	**Global WRF**  gWRF_merc	**Idealized cases**  wrf_Grav2x wrf_Hill2d wrf_Squall_2d_x wrf_Squall_2d_y wrf_Seabreeze2x wrf_BWave wrf_QSS

Visualizing WRF with gsn_csm_xxx scripts

Why?

- More control for customizing plots
- Don't want all those titles
- Plot WRF data on different map projection than what's on WRF file
- Can plot subset of data easier
- Need to compare with plots of non WRF data

Visualizing WRF with gsn_csm_xxx scripts

To plot data in NATIVE WRF map projection defined on file:

1. Call "`wrf_map_resources`" to set up map resources
2. Set `tfDoNDCOverlay` resource to True
3. Set `gsnAddCyclic` resource to False
4. Call one of the gsn_csm_xxx_map functions:
   - `gsn_csm_contour_map`
   - `gsn_csm_vector_map`
   - `gsn_csm_streamline_map`

Demo: plotting WRF using gsn_csm functions

GSN scripts with different levels of customization

`wrf_demo_plot_tc_gsn.ncl`

`wrf_demo_plot_tc_gsn_minor_custom.ncl`

`wrf_demo_plot_tc_gsn_major_custom.ncl`

Original WRF-NCL script (for comparison)

`wrf_demo_plot_tc.ncl`

gsn_csm script – native projection

```
a  = addfile("wrfout_d01_2005-08-28_00:00:00","r")
tc = wrf_user_getvar(a,"tc",0)

wks = gsn_open_wks("x11","wrf_demo_plot_tc_gsn")

;---Required to properly set up WRF map projection
 res = wrf_map_resources(a,True)
 res@tfDoNDCOverlay = True
 res@gsnAddCyclic   = False

 res@cnFillOn       = True
 res@cnLinesOn      = False

 plot = gsn_csm_contour_map(wks,tc(0,:,:),res)
```

Original WRF script

```
a  = addfile("wrfout_d01_2005-08-28_00:00:00","r")
tc = wrf_user_getvar(a,"tc",0)    ; Temperature (C)

wks = gsn_open_wks("x11","wrf_demo_plot_tc")

res           = True
res@cnFillOn = True

contour = wrf_contour(a,wks,tc(0,:,:),res)
plot    = wrf_map_overlays(a,wks,contour,False,False)
```

WRF-NCL script

gsn_csm script

Init: 2005-08-28_00:00:00





OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

Visualizing WRF with gsn_csm_xxx scripts

To plot data in NON-NATIVE map projection:

1. Set special "lat2d" / "lon2d" attributes
2. Set options for the map projection you want
3. Set `gsnAddCyclic` resource to False
4. Call one of the gsn_csm_xxx_map functions:
   - `gsn_csm_contour_map`
   - `gsn_csm_vector_map`
   - `gsn_csm_streamline_map`

gsn_csm script – non-native projection

```
a  = addfile("wrfout_d01_2005-08-28_00:00:00","r")
tc = wrf_user_getvar(a,"tc",0)

;---Required for using different map projection
 tc@lat2d = wrf_user_getvar(a,"lat",0)
 tc@lon2d = wrf_user_getvar(a,"lon",0)

wks = gsn_open_wks("x11","wrf_demo_plot_tc_gsn_nn")

 res@mpMinLatF       = min(tc@lat2d)-5  ; Select area of
 res@mpMaxLatF       = max(tc@lat2d)+5  ; map to view.
 res@mpMinLonF       = min(tc@lon2d)-5
 res@mpMaxLonF       = max(tc@lon2d)+5
 res@mpOutlineBoundarySets = "National"

 res@gsnAddCyclic    = False

 res@cnFillOn        = True
 res@cnLinesOn       = False

 plot = gsn_csm_contour_map(wks,tc(0,:,:),res)
```


Temperature
C
40N
30N
20N
100W
90W
80W
14
16
18
20
22
24
26
28
30
32
34
36

Major customization – Satellite map projection

wrf_demo_plot_tc_gsn_major_custom.ncl



NCL has support for shapefiles, allowing you to use the numerous free shapefiles for adding your own map outlines







OUTPUT FROM WRF V3.3 MODEL
WE = 225 ; SN = 175 ; Levels = 28 ; Dis = 8km ; Phys Opt = 6 ; PBL Opt = 1 ; Cu Opt = 1

Shapefiles give you detailed geographical outlines



Global Administrative Areas database (**http://www.gadm.org**) offers consistent administrative boundaries at many levels. The level 0 database (nations) is good to use for global or mesoscale results, level 1 is the first level of sub-national administration (typically states/provinces and territories) while level 2 offers the second level of administration and is potentially useful for high-resolution plots.

China shapefiles from gadm.org/country



Demo: adding shapefile outlines

Shapefiles downloaded from

http://www.gadm.org/country/

`wrf_demo_plot_tc_shapefiles.ncl`

`wrf_demo_plot_tc_gsn_shapefiles.ncl`

wrf_demo_plot_tc_shapefiles.ncl

```
f = addfile("wrfout_d01_2008-09-29_16:30:00","r")
tc = wrf_user_getvar(f,"tc",0)

wks = gsn_open_wks("x11","wrf_demo_plot_tc_shapefiles")

res = True
res@cnFillOn = True                      ; Turn on color fill
contour = wrf_contour(f,wks,tc(0,:,:),res)

;--- Create contours over a map (nothing is drawn because PanelPlot is True)
pltres = True
pltres@PanelPlot = True ; Tells wrf_map_overlays to not draw plot
pltres@NoTitles = True  ; Turn off titles
plot  = wrf_map_overlays(f,wks,contour,pltres,False)

;--- Add shapefile outlines
 lnres                  = True
lnres@gsLineColor      = "Gray25"
lnres@gsLineThicknessF = 3.          ; default is a little thin
usa_id = gsn_add_shapefile_polylines (wks,plot,"USA_adm/USA_adm0.shp",lnres)
mex_id = gsn_add_shapefile_polylines (wks,plot,"MEX_adm/MEX_adm1.shp",lnres)
cub_id = gsn_add_shapefile_polylines (wks,plot,"CUB_adm_shp/CUB_adm2.shp",lnres)

draw(plot)    ; Now draw the plot
frame(wks)    ; and advance the frame
```

USA, Mexico, Cuba shapefile outlines added





OUTPUT FROM WRF V3.7 MODEL
WE = 91 ; SN = 74 ; Levels = 30 ; Dis = 30km ; Phys Opt = 3 ; PBL Opt = 1 ; Cu Opt = 1

Demo: masking against shapefile outlines

`wrf_demo_plot_tc_gsn_shapefiles_mask.ncl`

`wrf_demo_plot_tc_shapefiles_mask.ncl`

```
load "./shapefile_utils.ncl"
tc = wrf_user_getvar(f,"tc",0)
tc@lat2d = wrf_user_getvar(f,"lat",0)    ; Needed for masking
tc@lon2d = wrf_user_getvar(f,"lon",0)

;--- Mask one level of tc against country outline of Cuba
 nl = 3      ; level index
 tc_mask = shapefile_mask_data(tc(nl,:,:),"CUB_adm0.shp", True)
```

Shapefiles downloaded from

http://www.gadm.org/country/


Temperature
C
36°N
34°N
32°N
30°N
28°N
26°N
24°N
22°N
20°N
100°W
95°W
90°W
85°W
80°W
24
24.4
24.8
25.2
25.6
26
26.4
26.8
27.2
27.6
28
28.4


Temperature
C
28.5
28
27.5
27
26.5
26
25.5
25
24.5
24
20N
85W
80W

NCL debugging tips

- Start with an existing script, if possible
- Use editor enhancements for coloring of syntax, functions, etc
- Use indentation (even though not needed)
- Use printVarSummary, printMinMax, print to examine variables
- Carefully read documentation for functions
- Read errors and warnings carefully ☺

Tips for graphics

- Make sure spelling the resource name correctly
- For nice-looking graphics:
  - Increase line thicknesses
  - Use color wisely
  - Use "SVG" format for web
  - For presentations: increase resolution of PNG images (use "`convert -trim `" to trim the images)

```
wtype           = "png"
wtype@wkWidth   = 2500
wtype@wkHeight = 2500
wks             = gsn_open_wks(wtype,"myplot")
```

Tips for new and advanced users

- Read the NCL User Guide
- Visit the NCL Examples page
- Join the ncl-talk email list
- Install a UNIX editor enhancement for NCL

Thank you!

http://www.ncl.ucar.edu/Training/Tutorials/WRF_Users_Workshop/